

ivprobit — Probit model with continuous endogenous covariates[Description](#)[Syntax](#)[Remarks and examples](#)[Acknowledgments](#)[Quick start](#)[Options for ML estimator](#)[Stored results](#)[References](#)[Menu](#)[Options for two-step estimator](#)[Methods and formulas](#)[Also see](#)

Description

`ivprobit` fits models for binary dependent variables where one or more of the covariates are endogenous and errors are normally distributed. By default, `ivprobit` uses maximum likelihood, but Newey's (1987) minimum chi-squared (two-step) estimator can be requested. Both estimators assume that the endogenous covariates are continuous and so are not appropriate for use with discrete endogenous covariates.

Quick start

Probit regression of y_1 on x and endogenous regressor y_2 that is instrumented using z

```
ivprobit y1 x (y2 = z)
```

With robust standard errors

```
ivprobit y1 x (y2 = z), vce(robust)
```

Use Newey's two-step estimator

```
ivprobit y1 x (y2 = z), twostep
```

As above, and show first-stage regression results

```
ivprobit y1 x (y2 = z), twostep first
```

Menu

Statistics > Endogenous covariates > Probit model with endogenous covariates

Syntax

Maximum likelihood estimator

```
ivprobit depvar [varlist1] (varlist2 = varlistiv) [if] [in] [weight] [, mle_options]
```

Two-step estimator

```
ivprobit depvar [varlist1] (varlist2 = varlistiv) [if] [in] [weight], twostep  
[tse_options]
```

*varlist*₁ is the list of exogenous variables.

*varlist*₂ is the list of endogenous variables.

*varlist*_{iv} is the list of exogenous variables used with *varlist*₁ as instruments for *varlist*₂.

mle_options

Description

Model

<u>m</u> le	use conditional maximum-likelihood estimator; the default
as <u>i</u> s	retain perfect predictor variables
<u>c</u> onstraints(<i>constraints</i>)	apply specified linear constraints

SE/Robust

vce(<i>vcetype</i>)	<i>vcetype</i> may be oim, <u>r</u> obust, <u>c</u> luster <i>clustvar</i> , opg, <u>b</u> ootstrap, or <u>j</u> ackknife
-----------------------	--

Reporting

<u>l</u> evel(#)	set confidence level; default is level(95)
<u>f</u> irst	report first-stage regression
<u>n</u> ocns <u>r</u> eport	do not display constraints
<i>display_options</i>	control columns and column formats, row spacing, line width, display of omitted variables and base and empty cells, and factor-variable labeling

Maximization

<i>maximize_options</i>	control the maximization process
<u>c</u> oef <u>l</u> egend	display legend instead of statistics

<i>tse_options</i>	Description
Model	
* twostep	use Newey's two-step estimator; the default is <code>mle</code>
asis	retain perfect predictor variables
SE	
vce (<i>vcetype</i>)	<i>vcetype</i> may be <code>twostep</code> , <code>bootstrap</code> , or <code>jackknife</code>
Reporting	
level (#)	set confidence level; default is <code>level(95)</code>
first	report first-stage regression
<i>display_options</i>	control columns and column formats, row spacing, line width, display of omitted variables and base and empty cells, and factor-variable labeling
coeflegend	display legend instead of statistics
* <code>twostep</code> is required.	

*varlist*₁ and *varlist*_{iv} may contain factor variables; see [U] 11.4.3 [Factor variables](#).

devar, *varlist*₁, *varlist*₂, and *varlist*_{iv} may contain time-series operators; see [U] 11.4.4 [Time-series varlists](#).

`bootstrap`, `by`, `jackknife`, `rolling`, `statsby`, and `svy` are allowed; see [U] 11.1.10 [Prefix commands](#). `fp` is allowed with the maximum likelihood estimator.

Weights are not allowed with the `bootstrap` prefix; see [R] [bootstrap](#).

`vce()`, `first`, `twostep`, and `weights` are not allowed with the `svy` prefix; see [SVY] [svy](#).

`fweights`, `iweights`, and `pweights` are allowed with the maximum likelihood estimator. `fweights` are allowed with Newey's two-step estimator. See [U] 11.1.6 [weight](#).

`coeflegend` does not appear in the dialog box.

See [U] 20 [Estimation and postestimation commands](#) for more capabilities of estimation commands.

Options for ML estimator

Model

`mle` requests that the conditional maximum-likelihood estimator be used. This is the default.

`asis` requests that all specified variables and observations be retained in the maximization process.

This option is typically not used and may introduce numerical instability. Normally, `ivprobit` drops any endogenous or exogenous variables that perfectly predict success or failure in the dependent variable. The associated observations are also dropped. For more information, see [Model identification](#) in [R] [probit](#).

`constraints`(*constraints*); see [R] [estimation options](#).

SE/Robust

`vce`(*vcetype*) specifies the type of standard error reported, which includes types that are derived from asymptotic theory (`oim`, `opg`), that are robust to some kinds of misspecification (`robust`), that allow for intragroup correlation (`cluster` *clustvar*), and that use bootstrap or jackknife methods (`bootstrap`, `jackknife`); see [R] [vce_option](#).

Reporting

`level(#)`; see [R] [estimation options](#).

`first` requests that the parameters for the reduced-form equations showing the relationships between the endogenous variables and instruments be displayed. For the two-step estimator, `first` shows the first-stage regressions. For the maximum likelihood estimator, these parameters are estimated jointly with the parameters of the probit equation. The default is not to show these parameter estimates.

`nocnsreport`; see [R] [estimation options](#).

`display_options`: `noci`, `nopvalues`, `noomitted`, `vsquish`, `noemptycells`, `baselevels`, `allbaselevels`, `nofvlabel`, `fvwrap(#)`, `fvwrapon(style)`, `cformat(%fmt)`, `pformat(%fmt)`, `sformat(%fmt)`, and `nolstretch`; see [R] [estimation options](#).

Maximization

`maximize_options`: `difficult`, `technique(algorithm_spec)`, `iterate(#)`, `[no]log`, `trace`, `gradient`, `showstep`, `hessian`, `showtolerance`, `tolerance(#)`, `ltolerance(#)`, `nrtolerance(#)`, `nonrtolerance`, and `from(init_specs)`; see [R] [maximize](#).

The following option is available with `ivprobit` but is not shown in the dialog box:

`coeflegend`; see [R] [estimation options](#).

Options for two-step estimator

Model

`twostep` is required and requests that Newey's (1987) efficient two-step estimator be used to obtain the coefficient estimates.

`asis` requests that all specified variables and observations be retained in the maximization process. This option is typically not used and may introduce numerical instability. Normally, `ivprobit` drops any endogenous or exogenous variables that perfectly predict success or failure in the dependent variable. The associated observations are also dropped. For more information, see [Model identification](#) in [R] [probit](#).

SE

`vce(vctype)` specifies the type of standard error reported, which includes types that are derived from asymptotic theory (`twostep`) and that use bootstrap or jackknife methods (`bootstrap`, `jackknife`); see [R] [vce_option](#).

Reporting

`level(#)`; see [R] [estimation options](#).

`first` requests that the parameters for the reduced-form equations showing the relationships between the endogenous variables and instruments be displayed. For the two-step estimator, `first` shows the first-stage regressions. For the maximum likelihood estimator, these parameters are estimated jointly with the parameters of the probit equation. The default is not to show these parameter estimates.

display_options: `noci`, `nopvalues`, `noomitted`, `vsquish`, `noemptycells`, `baselevels`, `allbaselevels`, `nofvlabel`, `fvwrap(#)`, `fvwrapon(style)`, `cformat(%fmt)`, `pformat(%fmt)`, `sformat(%fmt)`, and `nolstretch`; see [R] [estimation options](#).

The following option is available with `ivprobit` but is not shown in the dialog box:

`coeflegend`; see [R] [estimation options](#).

Remarks and examples

[stata.com](http://www.stata.com)

Remarks are presented under the following headings:

[Model setup](#)

[Model identification](#)

Model setup

`ivprobit` fits models with dichotomous dependent variables and endogenous covariates. You can use it to fit a probit model when you suspect that one or more of the covariates are correlated with the error term. `ivprobit` is to probit modeling what `ivregress` is to linear regression analysis; see [R] [ivregress](#) for more information.

Formally, the model is

$$\begin{aligned}y_{1i}^* &= \mathbf{y}_{2i}\boldsymbol{\beta} + \mathbf{x}_{1i}\boldsymbol{\gamma} + u_i \\ \mathbf{y}_{2i} &= \mathbf{x}_{1i}\boldsymbol{\Pi}_1 + \mathbf{x}_{2i}\boldsymbol{\Pi}_2 + \mathbf{v}_i\end{aligned}$$

where $i = 1, \dots, N$, \mathbf{y}_{2i} is a $1 \times p$ vector of endogenous variables, \mathbf{x}_{1i} is a $1 \times k_1$ vector of exogenous variables, \mathbf{x}_{2i} is a $1 \times k_2$ vector of additional instruments, and the equation for \mathbf{y}_{2i} is written in reduced form. By assumption, $(u_i, \mathbf{v}_i) \sim N(\mathbf{0}, \boldsymbol{\Sigma})$, where σ_{11} is normalized to one to identify the model. $\boldsymbol{\beta}$ and $\boldsymbol{\gamma}$ are vectors of structural parameters, and $\boldsymbol{\Pi}_1$ and $\boldsymbol{\Pi}_2$ are matrices of reduced-form parameters. This is a recursive model: \mathbf{y}_{2i} appears in the equation for y_{1i}^* , but y_{1i}^* does not appear in the equation for \mathbf{y}_{2i} . We do not observe y_{1i}^* ; instead, we observe

$$y_{1i} = \begin{cases} 0 & y_{1i}^* < 0 \\ 1 & y_{1i}^* \geq 0 \end{cases}$$

The order condition for identification of the structural parameters requires that $k_2 \geq p$. Presumably, $\boldsymbol{\Sigma}$ is not block diagonal between u_i and \mathbf{v}_i ; otherwise, \mathbf{y}_{2i} would not be endogenous.

□ Technical note

This model is derived under the assumption that (u_i, \mathbf{v}_i) is independent and identically distributed multivariate normal for all i . The `vce(cluster clustvar)` option can be used to control for a lack of independence. As with most probit models, if u_i is heteroskedastic, point estimates will be inconsistent. □

▷ Example 1

We have hypothetical data on 500 two-parent households, and we wish to model whether the woman is employed. We have a variable, `fem_work`, that is equal to one if she has a job and zero otherwise. Her decision to work is a function of the number of children at home (`kids`), number of years of schooling completed (`fem_educ`), and other household income measured in thousands of dollars (`other_inc`). We suspect that unobservable shocks affecting the woman's decision to hold a job also affect the household's other income. Therefore, we treat `other_inc` as endogenous. As an instrument, we use the number of years of schooling completed by the man (`male_educ`).

The syntax for specifying the exogenous, endogenous, and instrumental variables is identical to that used in `ivregress`; see [R] [ivregress](#) for details.

```
. use http://www.stata-press.com/data/r15/laborsup
. ivprobit fem_work fem_educ kids (other_inc = male_educ)

Fitting exogenous probit model
Iteration 0:  log likelihood = -344.63508
Iteration 1:  log likelihood = -252.10819
Iteration 2:  log likelihood = -252.04529
Iteration 3:  log likelihood = -252.04529

Fitting full model
Iteration 0:  log likelihood = -2368.2142
Iteration 1:  log likelihood = -2368.2062
Iteration 2:  log likelihood = -2368.2062

Probit model with endogenous regressors          Number of obs   =           500
Wald chi2(3)                                    =          163.88
Log likelihood = -2368.2062                      Prob > chi2     =           0.0000
```

	Coef.	Std. Err.	z	P> z	[95% Conf. Interval]	
other_inc	-.0542756	.0060854	-8.92	0.000	-.0662028	-.0423485
fem_educ	.2111111	.0268648	7.86	0.000	.1584569	.2637651
kids	-.1820929	.0478267	-3.81	0.000	-.2758315	-.0883542
_cons	.3672086	.4480724	0.82	0.412	-.5109971	1.245414
corr(e.other_~c, e.fem_work)	.3720375	.1300518			.0946562	.5958136
sd(e.other_~c)	16.66621	.5270318			15.66461	17.73186

```
Instrumented:  other_inc
Instruments:   fem_educ kids male_educ

Wald test of exogeneity (corr = 0):  chi2(1) = 6.70          Prob > chi2 = 0.0096
```

`ivprobit` used the default maximum likelihood estimator. The header of the output contains the sample size as well as a Wald statistic and p -value for the test of the hypothesis that all the slope coefficients are jointly zero. Below the table of coefficients, Stata reminds us that the endogenous variable is `other_inc` and that `fem_educ`, `kids`, and `male_educ` were used as instruments.

At the bottom of the output is a Wald test of the exogeneity of the instrumented variables. We reject the null hypothesis of no endogeneity. If there is no endogeneity, a standard probit regression would be preferable (see [R] [probit](#)).

Below we fit our model with Newey's (1987) minimum chi-squared estimator using the `twostep` option.

▷ Example 2

Refitting our labor-supply model with the two-step estimator yields

```
. ivprobit fem_work fem_educ kids (other_inc = male_educ), twostep
Checking reduced-form model...
```

```
Two-step probit with endogenous regressors      Number of obs   =      500
                                                Wald chi2(3)    =     93.97
                                                Prob > chi2     =     0.0000
```

	Coef.	Std. Err.	z	P> z	[95% Conf. Interval]	
other_inc	-.058473	.0093364	-6.26	0.000	-.0767719	-.040174
fem_educ	.227437	.0281628	8.08	0.000	.1722389	.282635
kids	-.1961748	.0496323	-3.95	0.000	-.2934522	-.0988973
_cons	.3956061	.4982649	0.79	0.427	-.5809752	1.372187

```
Instrumented:  other_inc
Instruments:  fem_educ kids male_educ
```

```
Wald test of exogeneity: chi2(1) = 6.50          Prob > chi2 = 0.0108
```

All the coefficients have the same signs as their counterparts in the maximum likelihood model. The Wald test at the bottom of the output confirms our earlier finding of endogeneity.

◀

□ Technical note

In a standard probit model, the error is assumed to have a variance of one. In the probit model with endogenous covariates, we assume that (u_i, v_i) is multivariate normal with covariance matrix

$$\text{Var}(u_i, v_i) = \Sigma = \begin{bmatrix} 1 & \Sigma'_{21} \\ \Sigma_{21} & \Sigma_{22} \end{bmatrix}$$

From the properties of the multivariate normal distribution, it follows that $\text{Var}(u_i|v_i) = 1 - \Sigma'_{21} \Sigma_{22}^{-1} \Sigma_{21}$. Newey's estimator and other two-step probit estimators yield estimates of β/σ and γ/σ , where σ is the square root of $\text{Var}(u_i|v_i)$, instead of estimates of β and γ . Hence, we cannot directly compare the estimates obtained from Newey's estimator with those obtained from maximum likelihood, which estimate β , γ , and σ separately. See [Wooldridge \(2010, 585–594\)](#) for a discussion about the interpretation of the estimates and the computation of marginal effects of two-step probit estimators under endogeneity.

□

Model identification

As in the linear simultaneous-equation model, the order condition for identification requires that the number of excluded exogenous variables (that is, the additional instruments) be at least as great as the number of included endogenous variables. `ivprobit` checks this for you and issues an error message if the order condition is not met.

Like `probit`, `logit`, and `logistic`, `ivprobit` checks the exogenous and endogenous variables to see if any of them predict the outcome variable perfectly. It will then drop offending variables and observations and fit the model on the remaining data. Instruments that are perfect predictors do not affect estimation, so they are not checked. See *Model identification* in [R] `probit` for more information.

`ivprobit` will also occasionally display messages such as

```
Note: 4 failures and 0 successes completely determined.
```

For an explanation of this message, see [R] `logit`.

Stored results

`ivprobit`, `mle` stores the following in `e()`:

Scalars

<code>e(N)</code>	number of observations
<code>e(N_cds)</code>	number of completely determined successes
<code>e(N_cdf)</code>	number of completely determined failures
<code>e(k)</code>	number of parameters
<code>e(k_eq)</code>	number of equations in <code>e(b)</code>
<code>e(k_eq_model)</code>	number of equations in overall model test
<code>e(k_aux)</code>	number of auxiliary parameters
<code>e(k_dv)</code>	number of dependent variables
<code>e(df_m)</code>	model degrees of freedom
<code>e(ll)</code>	log likelihood
<code>e(N_clust)</code>	number of clusters
<code>e(endog_ct)</code>	number of endogenous covariates
<code>e(p)</code>	model Wald p -value
<code>e(p_exog)</code>	exogeneity test Wald p -value
<code>e(chi2)</code>	model Wald χ^2
<code>e(chi2_exog)</code>	Wald χ^2 test of exogeneity
<code>e(rank)</code>	rank of <code>e(V)</code>
<code>e(ic)</code>	number of iterations
<code>e(rc)</code>	return code
<code>e(converged)</code>	1 if converged, 0 otherwise

Macros

<code>e(cmd)</code>	<code>ivprobit</code>
<code>e(cmdline)</code>	command as typed
<code>e(depvar)</code>	name of dependent variable
<code>e(instd)</code>	instrumented variables
<code>e(insts)</code>	instruments
<code>e(wtype)</code>	weight type
<code>e(wexp)</code>	weight expression
<code>e(title)</code>	title in estimation output
<code>e(clustvar)</code>	name of cluster variable
<code>e(chi2type)</code>	Wald; type of model χ^2 test
<code>e(vce)</code>	<code>vctype</code> specified in <code>vce()</code>
<code>e(vcetype)</code>	title used to label Std. Err.
<code>e(asis)</code>	<code>asis</code> , if specified
<code>e(method)</code>	<code>ml</code>
<code>e(opt)</code>	type of optimization

e(which)	max or min; whether optimizer is to perform maximization or minimization
e(ml_method)	type of ml method
e(user)	name of likelihood-evaluator program
e(technique)	maximization technique
e(properties)	b V
e(estat_cmd)	program used to implement estat
e(predict)	program used to implement predict
e(footnote)	program used to implement the footnote display
e(marginsok)	predictions allowed by margins
e(marginsprop)	signals to the margins command
e(asbalanced)	factor variables fvset as asbalanced
e(asobserved)	factor variables fvset as asobserved

Matrices

e(b)	coefficient vector
e(Cns)	constraints matrix
e(rules)	information about perfect predictors
e(ilog)	iteration log (up to 20 iterations)
e(gradient)	gradient vector
e(Sigma)	$\hat{\Sigma}$
e(V)	variance-covariance matrix of the estimators
e(V_modelbased)	model-based variance

Functions

e(sample)	marks estimation sample
-----------	-------------------------

ivprobit, twostep stores the following in e():

Scalars

e(N)	number of observations
e(N_cds)	number of completely determined successes
e(N_cdf)	number of completely determined failures
e(df_m)	model degrees of freedom
e(df_exog)	degrees of freedom for χ^2 test of exogeneity
e(p)	model Wald p -value
e(p_exog)	exogeneity test Wald p -value
e(chi2)	model Wald χ^2
e(chi2_exog)	Wald χ^2 test of exogeneity
e(rank)	rank of e(V)

Macros

e(cmd)	ivprobit
e(cmdline)	command as typed
e(depvar)	name of dependent variable
e(instd)	instrumented variables
e(insts)	instruments
e(wtype)	weight type
e(wexp)	weight expression
e(chi2type)	Wald; type of model χ^2 test
e(vce)	vctype specified in vce()
e(asis)	asis, if specified
e(method)	twostep
e(properties)	b V
e(estat_cmd)	program used to implement estat
e(predict)	program used to implement predict
e(footnote)	program used to implement the footnote display
e(marginsok)	predictions allowed by margins
e(marginsprop)	signals to the margins command
e(asbalanced)	factor variables fvset as asbalanced
e(asobserved)	factor variables fvset as asobserved

Matrices	
<code>e(b)</code>	coefficient vector
<code>e(rules)</code>	information about perfect predictors
<code>e(V)</code>	variance–covariance matrix of the estimators
Functions	
<code>e(sample)</code>	marks estimation sample

Methods and formulas

Fitting limited-dependent variable models with endogenous covariates has received considerable attention in the econometrics literature. Building on the results of Amemiya (1978, 1979), Newey (1987) developed an efficient method of estimation that encompasses both Rivers and Vuong’s (1988) simultaneous-equations probit model and Smith and Blundell’s (1986) simultaneous-equations tobit model. An efficient alternative to two-step estimation, and `ivprobit`’s default, is to use maximum likelihood. For compactness, we write the model as

$$y_{1i}^* = z_i \delta + u_i \tag{1a}$$

$$y_{2i} = x_i \Pi + v_i \tag{1b}$$

where $z_i = (y_{2i}, x_{1i})$, $x_i = (x_{1i}, x_{2i})$, $\delta = (\beta', \gamma')'$, and $\Pi = (\Pi_1', \Pi_2')'$.

Deriving the likelihood function is straightforward because we can write the joint density $f(y_{1i}, y_{2i} | x_i)$ as $f(y_{1i} | y_{2i}, x_i) f(y_{2i} | x_i)$. When there is an endogenous regressor, the log likelihood for observation i is

$$\ln L_i = w_i \left[y_{1i} \ln \Phi(m_i) + (1 - y_{1i}) \ln \{1 - \Phi(m_i)\} + \ln \phi \left(\frac{y_{2i} - x_i \Pi}{\sigma} \right) - \ln \sigma \right]$$

where

$$m_i = \frac{z_i \delta + \rho (y_{2i} - x_i \Pi) / \sigma}{(1 - \rho^2)^{\frac{1}{2}}}$$

$\Phi(\cdot)$ and $\phi(\cdot)$ are the standard normal distribution and density functions, respectively; σ is the standard deviation of v_i ; ρ is the correlation coefficient between u_i and v_i ; and w_i is the weight for observation i or one if no weights were specified. Instead of estimating σ and ρ , we estimate $\ln \sigma$ and $\text{atanh } \rho$, where

$$\text{atanh } \rho = \frac{1}{2} \ln \left(\frac{1 + \rho}{1 - \rho} \right)$$

For multiple endogenous covariates, let

$$\text{Var}(u_i, v_i) = \Sigma = \begin{bmatrix} 1 & \Sigma'_{21} \\ \Sigma_{21} & \Sigma_{22} \end{bmatrix}$$

As in any probit model, we have imposed the normalization $\text{Var}(u_i) = 1$ to identify the model. The log likelihood for observation i is

$$\ln L_i = w_i \left[y_{1i} \ln \Phi(m_i) + (1 - y_{1i}) \ln \{1 - \Phi(m_i)\} + \ln f(y_{2i} | x_i) \right]$$

where

$$\ln f(\mathbf{y}_{2i} | \mathbf{x}_i) = -\frac{p}{2} \ln 2\pi - \frac{1}{2} \ln |\Sigma_{22}| - \frac{1}{2} (\mathbf{y}_{2i} - \mathbf{x}_i \mathbf{\Pi}) \Sigma_{22}^{-1} (\mathbf{y}_{2i} - \mathbf{x}_i \mathbf{\Pi})'$$

and

$$m_i = (1 - \Sigma'_{21} \Sigma_{22}^{-1} \Sigma_{21})^{-\frac{1}{2}} \{ \mathbf{z}_i \boldsymbol{\delta} + (\mathbf{y}_{2i} - \mathbf{x}_i \mathbf{\Pi}) \Sigma_{22}^{-1} \Sigma_{21} \}$$

With maximum likelihood estimation, this command supports the Huber/White/sandwich estimator of the variance and its clustered version using `vce(robust)` and `vce(cluster clustvar)`, respectively. See [P] `_robust`, particularly *Maximum likelihood estimators* and *Methods and formulas*.

The maximum likelihood version of `ivprobit` also supports estimation with survey data. For details on VCEs with survey data, see [SVY] [variance estimation](#).

The two-step estimates are obtained using Newey's (1987) minimum chi-squared estimator. The reduced-form equation for y_{1i}^* is

$$\begin{aligned} y_{1i}^* &= (\mathbf{x}_i \mathbf{\Pi} + \mathbf{v}_i) \boldsymbol{\beta} + \mathbf{x}_{1i} \boldsymbol{\gamma} + u_i \\ &= \mathbf{x}_i \boldsymbol{\alpha} + \mathbf{v}_i \boldsymbol{\beta} + u_i \\ &= \mathbf{x}_i \boldsymbol{\alpha} + \nu_i \end{aligned}$$

where $\nu_i = \mathbf{v}_i \boldsymbol{\beta} + u_i$. Because u_i and \mathbf{v}_i are jointly normal, ν_i is also normal. Note that

$$\boldsymbol{\alpha} = \begin{bmatrix} \mathbf{\Pi}_1 \\ \mathbf{\Pi}_2 \end{bmatrix} \boldsymbol{\beta} + \begin{bmatrix} \mathbf{I} \\ \mathbf{0} \end{bmatrix} \boldsymbol{\gamma} = D(\mathbf{\Pi}) \boldsymbol{\delta}$$

where $D(\mathbf{\Pi}) = (\mathbf{\Pi}, \mathbf{I}_1)$ and \mathbf{I}_1 is defined such that $\mathbf{x}_i \mathbf{I}_1 = \mathbf{x}_{1i}$. Letting $\widehat{\mathbf{z}}_i = (\mathbf{x}_i \widehat{\mathbf{\Pi}}, \mathbf{x}_{1i})$, $\widehat{\mathbf{z}}_i \boldsymbol{\delta} = \mathbf{x}_i D(\widehat{\mathbf{\Pi}}) \boldsymbol{\delta}$, where $D(\widehat{\mathbf{\Pi}}) = (\widehat{\mathbf{\Pi}}, \mathbf{I}_1)$. Thus one estimator of $\boldsymbol{\alpha}$ is $D(\widehat{\mathbf{\Pi}}) \boldsymbol{\delta}$; denote this estimator by $\widehat{D} \boldsymbol{\delta}$.

$\boldsymbol{\alpha}$ could also be estimated directly as the solution to

$$\max_{\boldsymbol{\alpha}, \boldsymbol{\lambda}} \sum_{i=1}^N l(y_{1i}, \mathbf{x}_i \boldsymbol{\alpha} + \widehat{\mathbf{v}}_i \boldsymbol{\lambda}) \quad (2)$$

where $l(\cdot)$ is the log likelihood for probit. Denote this estimator by $\widetilde{\boldsymbol{\alpha}}$. The inclusion of the $\widehat{\mathbf{v}}_i \boldsymbol{\lambda}$ term follows because the multivariate normality of (u_i, \mathbf{v}_i) implies that, conditional on \mathbf{y}_{2i} , the expected value of u_i is nonzero. Because \mathbf{v}_i is unobservable, the least-squares residuals from fitting (1b) are used.

Amemiya (1978) shows that the estimator of $\boldsymbol{\delta}$ defined by

$$\max_{\boldsymbol{\delta}} (\widetilde{\boldsymbol{\alpha}} - \widehat{D} \boldsymbol{\delta})' \widehat{\boldsymbol{\Omega}}^{-1} (\widetilde{\boldsymbol{\alpha}} - \widehat{D} \boldsymbol{\delta})$$

where $\widehat{\boldsymbol{\Omega}}$ is a consistent estimator of the covariance of $\sqrt{N}(\widetilde{\boldsymbol{\alpha}} - \widehat{D} \boldsymbol{\delta})$, is asymptotically efficient relative to all other estimators that minimize the distance between $\widetilde{\boldsymbol{\alpha}}$ and $D(\widehat{\mathbf{\Pi}}) \boldsymbol{\delta}$. Thus an efficient estimator of $\boldsymbol{\delta}$ is

$$\widehat{\boldsymbol{\delta}} = (\widehat{D}' \widehat{\boldsymbol{\Omega}}^{-1} \widehat{D})^{-1} \widehat{D}' \widehat{\boldsymbol{\Omega}}^{-1} \widetilde{\boldsymbol{\alpha}} \quad (3)$$

and

$$\text{Var}(\widehat{\boldsymbol{\delta}}) = (\widehat{D}' \widehat{\boldsymbol{\Omega}}^{-1} \widehat{D})^{-1} \quad (4)$$

To implement this estimator, we need $\widehat{\boldsymbol{\Omega}}^{-1}$.

Consider the two-step maximum likelihood estimator that results from first fitting (1b) by OLS and computing the residuals $\widehat{v}_i = \mathbf{y}_{2i} - \mathbf{x}_i \widehat{\Pi}$. The estimator is then obtained by solving

$$\max_{\delta, \lambda} \sum_{i=1}^N l(y_{1i}, \mathbf{z}_i \delta + \widehat{v}_i \lambda)$$

This is the two-step instrumental variables (2SIV) estimator proposed by Rivers and Vuong (1988), and its role will become apparent shortly.

From Proposition 5 of Newey (1987), $\sqrt{N}(\widehat{\alpha} - \widehat{D}\delta) \xrightarrow{d} N(\mathbf{0}, \Omega)$, where

$$\Omega = J_{\alpha\alpha}^{-1} + (\lambda - \beta)' \Sigma_{22} (\lambda - \beta) Q^{-1}$$

and $\Sigma_{22} = E\{\mathbf{v}_i' \mathbf{v}_i\}$. $J_{\alpha\alpha}^{-1}$ is simply the covariance matrix of $\widetilde{\alpha}$, ignoring that $\widehat{\Pi}$ is an estimated parameter matrix. Moreover, Newey shows that the covariance matrix from an OLS regression of $\mathbf{y}_{2i}(\widehat{\lambda} - \widehat{\beta})$ on \mathbf{x}_i is a consistent estimator of the second term. $\widehat{\lambda}$ can be obtained from solving (2), and the 2SIV estimator yields a consistent estimate, $\widehat{\beta}$.

Mechanically, estimation proceeds in several steps.

1. Each of the endogenous right-hand-side variables is regressed on all the exogenous variables, and the fitted values and residuals are calculated. The matrix $\widehat{D} = D(\widehat{\Pi})$ is assembled from the estimated coefficients.
2. `probit` is used to solve (2) and obtain $\widetilde{\alpha}$ and $\widehat{\lambda}$. The portion of the covariance matrix corresponding to α , $J_{\alpha\alpha}^{-1}$, is also saved.
3. The 2SIV estimator is evaluated, and the parameters $\widehat{\beta}$ corresponding to \mathbf{y}_{2i} are collected.
4. $\mathbf{y}_{2i}(\widehat{\lambda} - \widehat{\beta})$ is regressed on \mathbf{x}_i . The covariance matrix of the parameters from this regression is added to $J_{\alpha\alpha}^{-1}$, yielding $\widehat{\Omega}$.
5. Evaluating (3) and (4) yields the estimates $\widehat{\delta}$ and $\text{Var}(\widehat{\delta})$.
6. A Wald test of the null hypothesis $H_0 : \lambda = \mathbf{0}$, using the 2SIV estimates, serves as our test of exogeneity.

The two-step estimates are not directly comparable to those obtained from the maximum likelihood estimator or from `probit`. The argument is the same for Newey's efficient estimator as for Rivers and Vuong's (1988) 2SIV estimator, so we consider the simpler 2SIV estimator. From the properties of the normal distribution,

$$E(u_i | \mathbf{v}_i) = \mathbf{v}_i \Sigma_{22}^{-1} \Sigma_{21} \quad \text{and} \quad \text{Var}(u_i | \mathbf{v}_i) = 1 - \Sigma_{21}' \Sigma_{22}^{-1} \Sigma_{21}$$

We write u_i as $u_i = \mathbf{v}_i \Sigma_{22}^{-1} \Sigma_{21} + e_i = \mathbf{v}_i \lambda + e_i$, where $e_i \sim N(0, 1 - \rho^2)$, $\rho^2 = \Sigma_{21}' \Sigma_{22}^{-1} \Sigma_{21}$, and e_i is independent of \mathbf{v}_i . In the second stage of 2SIV, we use a probit regression to estimate the parameters of

$$y_{1i} = \mathbf{z}_i \delta + \mathbf{v}_i \lambda + e_i$$

Because \mathbf{v}_i is unobservable, we use the sample residuals from the first-stage regressions.

$$\Pr(y_{1i} = 1 | \mathbf{z}_i, \mathbf{v}_i) = \Pr(\mathbf{z}_i \delta + \mathbf{v}_i \lambda + e_i > 0 | \mathbf{z}_i, \mathbf{v}_i) = \Phi \left\{ (1 - \rho^2)^{-\frac{1}{2}} (\mathbf{z}_i \delta + \mathbf{v}_i \lambda) \right\}$$

Hence, as mentioned previously, 2SIV and Newey’s estimator do not estimate δ and λ but rather

$$\delta_\rho = \frac{1}{(1 - \rho^2)^{\frac{1}{2}}} \delta \quad \text{and} \quad \lambda_\rho = \frac{1}{(1 - \rho^2)^{\frac{1}{2}}} \lambda$$

Acknowledgments

The two-step estimator is based on the `probitiv` command written by Jonah Gelbach of the University of Pennsylvania Law School and the `ivprob` command written by Joe Harkness formerly with the Institute of Policy Studies at Johns Hopkins University.

References

- Amemiya, T. 1978. The estimation of a simultaneous equation generalized probit model. *Econometrica* 46: 1193–1205.
- . 1979. The estimation of a simultaneous-equation tobit model. *International Economic Review* 20: 169–181.
- Finlay, K., and L. M. Magnusson. 2009. Implementing weak-instrument robust tests for a general class of instrumental-variables models. *Stata Journal* 9: 398–421.
- Miranda, A., and S. Rabe-Hesketh. 2006. Maximum likelihood estimation of endogenous switching and sample selection models for binary, ordinal, and count variables. *Stata Journal* 6: 285–308.
- Newey, W. K. 1987. Efficient estimation of limited dependent variable models with endogenous explanatory variables. *Journal of Econometrics* 36: 231–250.
- Rivers, D., and Q. H. Vuong. 1988. Limited information estimators and exogeneity tests for simultaneous probit models. *Journal of Econometrics* 39: 347–366.
- Smith, R. J., and R. W. Blundell. 1986. An exogeneity test for the simultaneous equation tobit model with an application to labor supply. *Econometrica* 54: 679–685.
- Wooldridge, J. M. 2010. *Econometric Analysis of Cross Section and Panel Data*. 2nd ed. Cambridge, MA: MIT Press.

Also see

- [R] [ivprobit postestimation](#) — Postestimation tools for `ivprobit`
- [R] [gmm](#) — Generalized method of moments estimation
- [R] [ivregress](#) — Single-equation instrumental-variables regression
- [R] [ivtobit](#) — Tobit model with continuous endogenous covariates
- [R] [probit](#) — Probit regression
- [ERM] [eprobit](#) — Extended probit regression
- [SVY] [svy estimation](#) — Estimation commands for survey data
- [XT] [xtprobit](#) — Random-effects and population-averaged probit models
- [U] [20 Estimation and postestimation commands](#)